```
10      !  MULTIPLE CREEP TEST DATA ACQUISITION PROGRAM: "MultiCrp"
20      !
30      !  J. C. Gibeling ** April 25, 1983
40      !         revised ** January 06, 1989
41      !         revised ** December 10, 1999
50      !
60      !  Division of Materials Science and Engineering
70      !  Department of Mechanical Engineering
80      !  University of California
90      !  Davis, California 95616
100     !
110     !  This program is to be used for data acquisition from one to five
120     !  creep tests (tension or compression). It is designed to collect data
130     !  points at the chosen time interval for each test. The range of time
140     !  intervals which is available is from 2 sec to 60 min per point.
150     !
160     !  This program uses BASIC 3.0, 4.0 or 5.0 on an HP 9000 BASIC Language
170     !  Proceseor with an HP 3497A Data Acquisition and Control Unit and an
180     !  internal hard drive. An external printer is optional. The following
190     !  binary programs are required:
200     !
210     !                  GRAPH                    ERR
220     !                  IO                       DISC
230     !                  TRANS                    CS80
240     !                  MAT                      HPIB
250     !                  KBD                      SERIAL
260     !                  CLOCK                    CRTB
270     !                  MS                       CRTA
280     !
290     !
300     OPTION BASE 1
310     COM /Data/ Load(5,500),Elong(5,500),Time(5,500),Temp(5,500),Status(5),Tot
al_temp(5)
320     COM /Data/ Material$(5)[20],Sample_number$(5)[7],INTEGER Test_type(5),Poi
nt_counter(5),Number_points(5)
330     COM /Paths/ @Data_acq_unit
340     COM /Flags/ INTEGER Display_test,Test_running(5),Data_transfer(5),High_sp
d_flag(5),Control_flag(5),Fast_points(5),End_flag(5),Plot_flag
350     COM /Disc/ Store_device$[8],Drive$[5],Primary_label$[6],Backup_label$[6],
File_name$[10],Parm_file$(5)[10]
360     COM /Disc/ Check$[1],INTEGER File(5),Switch_drive(5),Primary_space,Backup
_space,Full_flag
370     COM /Display1/ Time_origin(5),Strain_origin(5),Time_span(5),Strain_span(5
)
380     COM /Parms/ Area(5),Gage_length(5),Elong_cal(5),Load_cal(5),Start(5),Fini
sh(5),Elong_0(5),Tc_cal(5),Max_load(5),Max_elong(5),Nom_load(5)
381     COM /Parms/ Comment$(5)[150],Computer$[8],INTEGER Test_frame
400     COM /Timing/ INTEGER Interval(5),Data_interval,Interval_flag
401     COM /Tcparm/ Parms(0:8,1:5)
410     DIM Date$[11],Time$[8],Response$[20],Print_device$[5]
420     INTEGER I,J,Point,Check,Ser_poll
430     !
440     !
450     GRAPHICS OFF
460     DEG
470     CONTROL 2,1;0                          ! Cancel PRINTALL mode
480     CONTROL 1,4;0                          ! Cancel DISPLAY FUNCTIONS mode
490     CONTROL 2,2;1                          ! Turn on USER softkeys
500     Computer$=SYSTEM$("SYSTEM ID")         ! Determine computer model
510     DUMP DEVICE IS 26                      ! Parallel printer
```

```
520     !ON TIMEOUT 26,.1 GOTO No_ext_printer
530     !Ser_poll=SPOLL(701)
540     !OUTPUT 701;CHR$(27)&"*rK"              ! Request printer model number
550     !ENTER 701 USING "5A";Print_device$
560     !
570 No_ext_printer:   !
580     !
590     !OFF TIMEOUT
591     !RESET 7
600     !SELECT Print_device$
610       !CASE "2673A"
620         !OUTPUT 701;CHR$(27)&"*r90X"        ! Horizontal graphics centering
630       !CASE ""
640         !IF Ser_poll<>0 THEN
650           !Print_device$="2225A"            ! Assume HP 2225A printer
660           !OUTPUT 701;CHR$(27)&"&l1L"       ! Perforation skip
670           !OUTPUT 701;CHR$(27)&"&s0C"       ! End-of-line wrap-around
680         !END IF
690     !END SELECT
700     PRINTER IS 1                            ! CRT
710     MASS STORAGE IS "\BLP\DATA:DOS,C"       ! Use system disc for data
720     OUTPUT 2 USING "#,B";255,75             ! [CLR SCR]
730     !
740     !
750     PRINT TABXY(15,4);"MULTIPLE CREEP TEST DATA ACQUISITION PROGRAM"
760     FOR I=1 TO 5
770       Status(I)=0                           ! Clear test status signal values
780       Test_running(I)=0                     ! Clear test running flags
790       Data_transfer(I)=1                    ! Set data transferred flags
800       High_spd_flag(I)=0                    ! Clear high speed data flag
810       End_flag(I)=0                         ! Clear end-of-test flag
820     NEXT I
830     !
840     Interval_flag=0                         ! Clear data interval flag
850     Plot_flag=1                             ! Enable data plotting
860     Display_test=1                          ! Default is to display data from l
oad frame 1
870     !
880     !ON ERROR GOTO 910
890     !DELSUB Phyread                         ! Delete previous disc utilities
900     !DELSUB Phywrite
910     !OFF ERROR
920     !LOADSUB ALL FROM "PHYREC"
921     !
923     !DISP USING "K";"Please insert the system data disc in the left hand driv
e, then press CONTINUE."
924     !CONTROL 2,2;0                          ! Turn on SYSTEM softkeys
925     !PAUSE
926     CONTROL 2,2;1                           ! Turn on USER softkeys
927     OUTPUT 2 USING "#,B";255,75             ! [CLR SCR]
930     !
940     !
950     CALL Set_clock
960     !
970     !
980     ! SET-UP 3497A DATA ACQUISITION/CONTROL UNIT
990     !
1000    ASSIGN @Data_acq_unit TO 709
1010    ON TIMEOUT 7,.1 CALL No_acq_unit
1020    Ser_poll=SPOLL(@Data_acq_unit)
```

```
1030    OFF TIMEOUT
1040    CLEAR @Data_acq_unit
1050    RESET 7
1060    !
1070    OUTPUT @Data_acq_unit USING "K";"VT4VD5VR3VA1VF1VN20VS1AF0AL19AC0AE1SO1SE
001"
1080                                          ! VT4: Hold DVM trigger
1090                                          ! VD5: 5-1/2 digit DVM output
1100                                          ! VR3: 10 Volt DVM range
1110                                          ! VA1: DVM Auto-zero on
1120                                          ! VF1: ASCII output format
1130                                          ! VN20: 20 readings per trigger
1140                                          ! VS1: DVM storage on
1150                                          ! AF0: First channel is 0
1160                                          ! AL19: Last channel is 19
1170                                          ! AC0: Close channel 0
1180                                          ! AE1: Enable external increment
1190                                          ! SO1: System output wait on
1200                                          ! SE001: Data ready SRQ mask
1201    OUTPUT @Data_acq_unit;"VC3"           ! 1 mA current for status signals
1210    !
1220    !
1230    CALL Test_setup
1240    !
1250    OUTPUT 2 USING "#,B";255,75           ! [CLR SCR]
1260    ON INTR 7,5 CALL Data                 ! Set-up SRQ service routine
1270    ENABLE INTR 7;2                       ! Enable SRQ interrupt on HPIB
1280    !
1290    OFF KEY
1300    SELECT Computer$[1,4]
1310      CASE "9817","S300","PC30"
1320        ON KEY 1 LABEL " Select   plot" CALL Select_plot
1330        ON KEY 2 LABEL "New data  rate" CALL Interval
1340        ON KEY 3 LABEL "  New     spans" CALL New_spans
1350        !ON KEY 4 LABEL "  Data  transfer" CALL Transfer
1360        ON KEY 5 LABEL " Start    test" CALL Test_setup
1370        ON KEY 6 LABEL "End test" CALL End_of_test
1380        ON KEY 8 LABEL "End all  tests" GOSUB End_all
1390      CASE ELSE
1400        ON KEY 0 LABEL "Select plot" CALL Select_plot
1410        ON KEY 1 LABEL "New data rate" CALL Interval
1420        ON KEY 2 LABEL "New spans" CALL New_spans
1430        !ON KEY 3 LABEL "Data transfer" CALL Transfer
1440        ON KEY 4 LABEL "Start test" CALL Test_setup
1450        ON KEY 5 LABEL "End test" CALL End_of_test
1460        ON KEY 9 LABEL "End all tests" GOSUB End_all
1470    END SELECT
1480    !
1490    !
1500    OUTPUT @Data_acq_unit;"VT2"           ! DVM in external trigger mode
1510    OUTPUT 2 USING "#,B";255,75           ! [CLR SCR]
1520    !
1530    GRAPHICS ON
1540    !
1550    !
1560  Idle3:                                  ! Main program loop
1570    Point=0
1580    IF Point_counter(Display_test)>1 THEN Point=Point_counter(Display_test)-1
1590    IF Point>0 THEN
1600      SYSTEM PRIORITY 2                   ! Raise priority to avoid ON KEY's
```

```
1610      !
1620      !
1630      ! INCREASE SPANS IF NECESSARY
1640      !
1650      IF ABS(Time(Display_test,Point)-Start(Display_test))/3600>ABS(Time_span
(Display_test)) THEN
1660        Time_span(Display_test)=2*Time_span(Display_test)
1670        CALL Strain_plot(Display_test)
1680      END IF
1690      !
1700      IF Test_type(Display_test)=1 THEN
1705        Strain=100*ABS(Elong_cal(Display_test)*(Elong(Display_test,Point)-Elo
ng_0(Display_test))/Gage_length(Display_test))
1710        IF Strain>ABS(Strain_span(Display_test))+Strain_origin(Display_test)
THEN
1720          Strain_span(Display_test)=2*Strain_span(Display_test)
1730          CALL Strain_plot(Display_test)
1740        END IF
1750      ELSE
1760        IF 100*ABS(Elong_cal(Display_test)*(Elong(Display_test,Point)-Elong_0(
Display_test)))>ABS(Strain_span(Display_test))+Strain_origin(Display_test) THEN
1770          Strain_span(Display_test)=2*Strain_span(Display_test)
1780          CALL Strain_plot(Display_test)
1790        END IF
1800      END IF
1810      SYSTEM PRIORITY 0                    ! Return to normal priority
1820    END IF
1830    !
1840    !
1850    ! END A TEST IF THAT STATUS SIGNAL IS ZERO
1860    !
1870    FOR I=1 TO 5
1880      IF Test_running(I) THEN
1890        Point=0
1900        IF Point_counter(I)>1 THEN Point=Point_counter(I)-1
1910        IF Point>0 THEN
1920          IF (ABS(Load(I,Point))>ABS(Max_load(I))) THEN Max_load(I)=Load(I,Po
int)
1930          IF (ABS(Elong(I,Point))>ABS(Max_elong(I))) THEN Max_elong(I)=Elong(
I,Point)
1940          IF ABS(Status(I))<.001 THEN End_flag(I)=1
1950          IF End_flag(I) THEN CALL End_of_test(I)
1960          !
1970          IF High_spd_flag(I)=1 THEN
1980            IF Point+1+(File(I)-1)*500>Fast_points(I) THEN
1990              High_spd_flag(I)=0           ! Clear high speed data flag
2000              Data_interval=Interval(I)
2010              FOR J=1 TO 5                 ! Find minimum data interval
2020                IF (Interval(J)>0 AND Interval(J)<Data_interval) THEN Data_in
terval=Interval(J)
2030              NEXT J
2040              Interval_flag=1              ! Output selected data interval
2050            END IF
2060          END IF
2070        END IF
2080      END IF
2090    NEXT I
2100    !
2110    IF Interval_flag THEN
2120      OUTPUT @Data_acq_unit USING "K,6Z";"VT2TI",100*(Data_interval DIV 60)+(
```

```
Data_interval MOD 60)
2130      Interval_flag=0
2140    END IF
2150    !
2160    !
2170    ! EXIT PROGRAM IF NO TESTS RUNNING
2180    !
2190    Check=0
2200    FOR I=1 TO 5
2210      Check=Check+Test_running(I)
2220    NEXT I
2230    IF NOT Check THEN GOSUB End_all
2240    !
2250    DISP                                        ! Clear display line
2260    GOTO Idle3                                  ! Loop to wait for data
2270    !
2280    !
2290    ! ****************** SUBROUTINES FOR MAIN PROGRAM **********************
2300    !
2310    !
2320  End_all: !
2330    !
2340    FOR I=1 TO 5
2350      IF Test_running(I) THEN
2360        INPUT "Do you wish to end all of the currently running tests? (Y/N)",
Response$
2370        IF UPC$(Response$[1,1])<>"Y" THEN
2380          RETURN
2390        ELSE
2400          GOTO 2460
2410        END IF
2420      END IF
2430    NEXT I
2440    !
2450    OFF INTR
2460    CLEAR 709
2470    !
2480    FOR I=1 TO 5
2490      IF Test_running(I) THEN CALL End_of_test(I)
2500    NEXT I
2510    !
2520    FOR I=1 TO 5                                ! Transfer data to floppy disc
2530      IF NOT Data_transfer(I) THEN CALL Transfer(I)
2540    NEXT I
2550    !
2551    PRINTER IS 1
2560    GRAPHICS OFF
2570    DISP "All tests have ended."
2580    END
2590    !
2600    !
2610    !********************* SUBPROGRAMS AND SUBFUNCTIONS*********************
2620    !
2630    !
2640   SUB Disc_setup2(INTEGER File_size)
2650       !
2660       ! J. C. Gibeling ** November 06, 1982
2670       !        revised ** July 19, 1988
2680       !
2690       ! This subroutine is used to set_up discs for test data storage.
```